
УДК 681.3.06 
 
КУЗНЕЦОВ А.В. 
 

СМЫСЛОВАЯ АМОРФНОСТЬ В ГРАММАТИЧЕСКИХ 
СИГНАТУРАХ ПРЕДЛОЖЕНИЙ ЕСТЕСТВЕННОГО ЯЗЫКА 

 
Рассмотрена технология анализа предложений естественного языка на основе 

грамматики непосредственных составляющих при раздельном представлении сло-
варя и грамматической сигнатуры. Описана структура модифицированного слова-
ря и смысловые агенты-процедуры, которые связывают лексемы анализируемых 
предложений с моделью проблемной области знаний и дополняют грамматиче-
ское согласование частей речи смысловым согласованием. Показано, что в данном 
случае грамматическая сигнатура теряет чувствительность к словарному напол-
нению и обретает свойство смысловой аморфности, сохраняя грамматический 
строй естественного языка. Такое свойство позволяет в рамках единой грам-
матической сигнатуры выполнять смысловой машинный анализ предложений с 
формально заданным стилем написания, но относящихся к разным проблемным 
областям, которые требуют отличающихся словарей. 

 
Consideration of technology natural language sentences computerized analysis 

based on definite clause grammar provided separate notation of dictionary and grammar 
signature is represented. There is describing modified dictionary and agents-procedures 
of meaning which associate the lexical elements inside sentence with problem area 
model and supplement the grammar harmonization with semantic harmonization 
between ones. In this case the grammar signature gets lost sensitivity to word's filling 
due that acquire amorphous semantics but keep natural language grammar order. Like 
distinctive feature inside uniform grammar signature provided the computerized 
analysis of meaning the sentences which have formal definite style for writing but 
associated with different problem knowledge areas that requires unlike dictionaries. 

 
В прикладной лингвистике широко используется грамматика непо-

средственных составляющих (ГНС), которая дает возможность структур-
ного описания синтаксических объектов предложений естественного язы-
ка на основе деревьев непосредственных составляющих, где ветви и ли-
стья ассоциируются с допустимыми списками лексем. В Прологе правиль-
ность использования форм таких лексем проверяется не преобразуемыми 
подцелями, которые заключены в фигурные скобки внутри граммати-
ческих правил и содержат необходимые атрибуты для согласования час-
тей речи в предложении. Другими словами, ГНС представляет собой од-
новременно и словарь и грамматику [1]. Подстановка вместо списков 
лексем правил обращения к словарю проблемной области (ПО) логически 
разделяет ГНС на две составные части – словарь и грамматическую сиг-
натуру, которая, таким образом, теряет чувствительность к словарному 
наполнению ГНС, но сохраняет грамматический строй выбранного есте-

ственного языка. Такая подстановка дает возможность в пределах единой 
грамматической сигнатуры (т.е. без изменения грамматических правил) 
обращаться к разным проблемным словарям одного и того же языка, а, 
значит, выполнять машинный анализ предложений, которые передают 
разный смысл или даже относятся к разным ПО, если выполнить необхо-
димые изменения в запрограммированной модели ПО. Т.е. отдельно 
представленная в ГНС грамматическая сигнатура приобретает свойство 
смысловой аморфности, и может использоваться как основа для организа-
ции обращений к различным моделям ПО на ограниченном естественном 
языке с формализованным стилем записи предложений. 

Следует учесть, что одна из существенных особенностей, касающихся 
исследования возможностей ГНС, сводится к иллюстрации отдельных 
примеров, так как ГНС уже представляет собой формализм для описания 
грамматики ограниченного естественного языка и не требует дополни-
тельных синтаксических надстроек для собственного описания. Таким 
образом, отображая обобщение ГНС на конструкции Пролога, определим 
подход к проектированию системы обработки естественного языка, ори-
ентированной на машинный анализ семантики заданного подмножества 
предложений, которые передают разный смысл, но могут быть пред-
ставлены единой грамматической сигнатурой. 

Абстрагируясь от внутренних разделителей и знаков препинания, схе-
матически зададим состав некоторой группы предложений в виде после-
довательности нетерминальных символов, перечисляющих части речи: 

 
<прилагатеное1><прилагательное2><существиительное1><глагол><существительное2>. 
 

Эта схема соответствует корневой вершине дерева разбора, которая в 
Прологе задается типовым определением контекстно-свободной грамма-
тики: 

 
'предложение'-->'фраза_существительного','фраза_глагола'. 

 
В одном из вариантов грамматического разбора такое разбиение пред-

ложения ассоциируется с двумя ветвями, исходящими из вершины перво-
го уровня иерархии дерева разбора, а 'фраза_существительного' потребует 
одного листа (для первого определителя существительного) и ветви, исхо-
дящей из вершины второго уровня, где представлен первый остаток спи-
ска лексем фразы существительного. 

Подобным же образом разбирается и 'фраза_глагола'.  
Далее, рассмотрим несколько предложений естественного языка: 
 

Смешная рыжая обезьяна держит банан. 
Знаменитый талантливый пародист рассмешил публику. 
Легкомысленный халатный охранник прозевал грабителя. 
Опасный вооруженный гангстер грабит банки. 

Очевидно, все эти предложения имеют общую характерную черту – 
они полностью вписываются в приведенную выше схему порядка следо-
вания частей речи естественного языка, хотя передают совершенно раз-
ный смысл. Для того чтобы в Прологе привести систему машинного ана-
лиза этих предложений к единой грамматической сигнатуре, остановимся, 
прежде всего, на организации словаря, вынесенного за пределы ГНС. 
Этот словарь потребует некоторой модификации по сравнению со слова-
рем, обычно встроенным в ГНС. 

В качестве основного требования к словарю определим общность 
структур, используемых для представления различных частей речи. Тогда 
любые факты, задающие требуемый словарь, могут быть записаны в Про-
логе следующим образом: 

 
<функтор>(<атрибут1>:<атрибут2>: ...:<атрибутm>,[<слово1>,<слово2>, ...,<словоn>]). 

 
В угловых скобках, как всегда в подобных случаях, представлены не-

терминальные символы. 
Для описания структурных составляющих словаря введем соглашение 

обозначать совпадающие по назначению атрибуты разных частей речи од-
ними и теми же большими буквами латиницы. В программе на Прологе с 
больших букв начинаются имена переменных, содержащих значение того 
или иного атрибута. Что касается представления самих значений исполь-
зуемых атрибутов, то они задаются атомами Пролога, которые ассоцииру-
ются с названиями этих значений на естественном языке, но с приемле-
мыми сокращениями, допускающими сохранение читабельности. Таким 
образом, в условиях ввода ограничений на естественный язык при по-
строении ГНС, будем использовать атрибуты, перечисленные в табл. 1., а 
их допустимые значения представлены в табл. 2. Тогда, в соответствии с 
данными из таблиц 1,2, для грамматического согласования частей речи в 
предложении будем использовать наборы атрибутов, которые сведены в 
табл. 3.  

Табл.1 
 Существи-

тельное 
Прилага-
тельное 

Место-
имение 

Глагол Параметр 
смыслового 
агента 

G – род × – × × – 
Q – число × × × × – 
A – падеж × × × – – 
R – разряд или 
лицо 

– 
 

– 
 

× × – 

T – время – – – × – 
V – вид – – – × – 
D – залог – – – × – 

 Существи-
тельное 

Прилага-
тельное 

Место-
имение 

Глагол Параметр 
смыслового 
агента 

C – класс при-
менимости 

× – – × – 

F – функция 
применимости 

– – – – × 

S – субъект 
действия 

– – – – × 

 
Табл. 2 

G Q A R T V 
'мужск' 
'женск' 
'средн' 
'неопфм' 
 
 

'единств' 
'мнжеств' 
'неопфм' 
 

'именит' 
'родит' 
'дател' 
'винит' 
'творит' 
'прелож' 

'личн'('первое' ) 
'личн'('второе') 
'личн'('третье') 
'безличн' 
'возвр' 
'притяж' 
'указ' 
'опред' 
'притяж' 
'указ' 
'опред' 
'вопрос' 
'относит' 
'отриц' 
'неопфм' 

'прош' 
'наст' 
'будущ' 
'неопф
м' 
 

'сверш' 
'несверш' 

 
Табл.2 (продолжение) 

D C F S 
'действит' 
'страдат' 
'возврзн' 
'взаимнзн' 
'неопфм' 

<классы> 
 

<функции> <субъекты> 
 

 
Табл. 3 

Существи- 
тельное 

Прилагательное Местоимение Глагол 

G:Q:A:C:GB G:Q:A G:Q:A:R G:Q:R:T:V:D:C 
 
Для значений разрядов местоимений и лиц глагола применяются следую-
щие сокращения: 
Л и ч н ы е: 'личн'( 'первое' )  – первое лицо глагола, 
             'личн'( 'второе'  )  – второе лицо глагола, 
             'личн'( 'третье' ) – третье лицо глагола, 

             'беличн'   – безличные глаголы. 
В о з в р а т н о е:   – 'возвр'. 
П р и т я ж а т е л ь н ы е: – 'притяж'. 
У к а з а т е л ь н ы е:         – 'указ'. 
О п р е д е л и т е л ь н ы е:   – 'опред'. 
В о п р о с и т е л ь н ы е: – 'вопрос'. 
О т н о с и т е л ь н ы е: – 'относит'. 
Н е о п р е д е л е н н ы е:     – 'неопфм'. 
О т р и ц а т е л ь н ы е:       – 'отриц'. 

 

В дальнейших примерах использована лишь часть лиц глаголов, но 
здесь перечислены все разряды и лица, чтобы показать возможность пол-
ноты представления. 

Теперь обратимся к самой сложной части речи – глаголу. 
У глагола, помимо типовых признаков – рода, числа и лица, характер-

ных для его связи с другими частями речи, присутствуют время, вид, 
залог, и класс применимости, для дополнительной смысловой связи с 
субъектом в предложении. 

Наличие у глагола неопределенной формы приводит к необходимости 
использовать значение 'неопфм' в атрибутах, определяющих время, чис-
ло, лицо, род и залог. Обратим также внимание, что при использовании 
личных местоимений с глаголами значение разряда местоимения и лица 
глагола должны совпадать (согласование по переменной  R). 

Переменная D применяется для хранения значений залога ('дейст-
вит' – действительный, 'страдат' – страдательный), неопределенной 
формы ('неопфм') и возможных дополнительных значений ('возврзн' –
 возвратное значение, 'взаимнзн' – взаимное значение ), которые зави-
сят от роли глагола в предложении. 

Переменная C определяет класс применимости глагола и обеспечивает 
смысловую связку с существительным, выражающим субъект действия в 
предложении. 

Для пояснения необходимости введения такого атрибута как класс 
применимости рассмотрим три предложения, совершенно корректные с 
грамматической точки зрения. 

 
Знаменитый клоун держит в доме смешную обезьяну. 
Смешная обезьяна держит банан. 
Смешная обезьяна читает книгу. 
 

Очевидно, первые два предложения с глаголом 'держит' следует счи-
тать семантически правильными, тогда как третье предложение совер-
шенно лишено смысла ввиду отсутствия в человеческой практике грамот-
ных обезьян. Это значит, что глагол 'держит' применим, например, как к 
классу 'люди' так и к классу 'животные', тогда как глагол 'читает' 

может быть использован только с классом 'люди'. В терминах Пролога 
применимость требуемого глагола к одному или нескольким классам 
описывается в словаре следующим образом. 
 
'перех_глаг'('мужск':'единств':'личн'('третье'):'наст':'несоверш':'действит':C:, 
   ['держит'],Lc) :- ['животные', 'люди']=Lc,'принадлежит'(C,Lc). 
'перех_глаг'('мужск':'единств':'личн'('третье'):'наст':'несоверш': 
   'действит':'люди',['читает'],['люди']). 
'принадлежит'(X,[ X|_ ]). 
'принадлежит'(X,[_|T]) :- 'принадлежит'(X,T). 
 

При задании словаря глаголов следует также обратить внимание на ис-
пользование функторов, которые описывают такие свойства глаголов как 
переходность и непереходность. Переходные глаголы имеют функтор 
'перех_глаг', а непереходные – 'неперех_глаг'. Для словарей других 
частей речи функторы не несут дополнительной смысловой нагрузки и 
обозначаются как 'словарь_существит' и 'словарь_прилагат'. 

В первом определении структуры с функтором 'перех_глаг' на мес-
те обозначения класса использована переменная C, которая в ходе анализа 
предложения унифицируется некоторым значением, соответствующим 
названию класса существительного (субъекта действия в предложении). 
Тогда смысловая правильность использования глагола 'держит' по от-
ношению к существительному будет проверена предикатом 'принадле-
жит', который просмотрит список классов применимости. Второе опре-
деление указывает на то, что глагол 'читает' может использоваться 
только с уникальным классом 'люди', подстановка названия которого 
сразу сделана на месте переменной C. 

Подобно классу применимости C, переменные F и S необходимы для 
смысловой проверки анализируемых лексем. Эти переменные распро-
страняют свои значения по цепочке определитель→субъект→гла-
гол→объект в ходе грамматического разбора предложения. Такая инфор-
мация передается друг другу смысловыми агентами-процедурами (далее, 
просто смысловые агенты), которые используют переменные F  и S  в 
качестве параметров по следующей схеме: 

 
'агент_о'(Ld,S)→'агент_с'(n,S,F)→'агент_ф'(Lv,F,Fo)→'агент_фо'(Lo,Fo). 

 
Предикаты с соответствующими функторами приведены далее в рас-

ширенном примере, содержащем полную грамматическую сигнатуру для 
разбора предложений, которые имеют предусмотренный в ГНС стиль 
написания. 

 В ходе грамматического разбора смысловые агенты обеспечивают 
связь с моделью ПО для выяснения правильного смыслового использова-
ния каждой лексемы (Ld, Ls, Lv, Lo) во взаимном окружении, путем про-
верки допустимых смысловых отношений составленных из их базовых 

форм. Смысловой агент, отвечающий за работу с текущей анализируемой 
лексемой, отыскивает в словаре ее базовую форму, и обращается с запро-
сом к той части модели ПО, которая охватывает множество сопутствую-
щих грамматическому разбору структур, попадающих под признаки связи 
с анализируемым контекстом. Эти структуры составляются на шаге про-
ектирования модели ПО и образуют допустимые смысловые отношения 
на множестве включенных в них атомов. Значения атомов соответствуют 
базовым формам лексем, охваченных словарем. Тогда, ответ на запрос 
смыслового агента подтвержден успехом в том случае, когда найден атом, 
значение которого в допустимом смысловом отношении эквивалентно 
лексеме, приведенной к базовой форме. Т.е. успешная работа смысловых 
агентов обеспечивает однозначную привязку лексем в грамматической 
структуре предложения к атомам сопутствующих структур модели ПО. 
Если после завершения грамматического разбора все лексемы предложе-
ния охвачены допустимыми связями с атомами сопутствующих структур, 
то разобранное предложение принимается, как семантически правильное, 
так как лексемы во взаимном окружении образуют допустимые смысло-
вые отношения в контексте заданных правил ГНС. 

Более детально алгоритм работы смысловых агентов выглядит сле-
дующим образом. 

Перед вызовом процедуры 'агент_о'(Ld,S)  формируется входной список 
лексем-определителей субъекта действия, который извлекается из анали-
зируемого предложения путем отсечения предусмотренного в ГНС коли-
чества элементов от головы списка лексем (в нашем случае – два элемен-
та). Полученные элементы приводятся к своей базовой форме. Далее, 
процедура 'агент_о'(Ld,S) через параметр Ld  принимает входной список 
лексем-определителей и при  помощи встроенного предиката bagof в базе 
знаний собирается информация о всех субъектах, которые попадают под 
заданные в определителях признаки. В результате формируется список 
структур, содержащих информацию о субъектах действия, для унифика-
ции с переменной S. Если этот список не пустой, то дальше его обрабаты-
вает 'агент_с'(Ls,S,F), отвечающий за работу с субъектами. 

Текущая лексема предложения Ls, в соответствии с грамматическими 
правилами, представляет собой субъект действия и, 'агент_с'(Ls,S,F) попы-
тается отыскать в списке S структуру с названием объекта, которое экви-
валентно базовой форме Ls. Если субъект действия найден, то 
'агент_с'(Ls,S,F) унифицирует переменную F  со списком функций, которые 
может выполнять найденный субъект. Затем, 'агент_ф'(Lv,F,Fo) обрабаты-
вает полученный от 'агент_с'(Ls,S,F) список F. 

Следующая лексема Lv представляет собой глагол, обозначающий 
действие, или, в нашей интерпретации, функцию применимости субъекта. 
Тогда 'агент_ф'(Lv,F,Fo) попытается отыскать в списке F такую функцию, 
функтор которой эквивалентен глаголу Lv в базовой форме. Если функция 
найдена, переменная Fo унифицируется со структурой, выражающей эту 
функцию, и завершение обработки выполняет 'агент_фо'(Lo,Fo). 

Процедура 'агент_фо'(Lo,Fo) попытается отыскать среди аргументов 
функции применимости такой, который эквивалентен базовой форме 
последней лексемы Lo, представляющей собой объект действия. Если 
эквивалентность установлена, то смысловой анализ предложения прини-
мается как успех. 

В расширенном примере проиллюстрирована работа всей цепочки 
смысловых агентов. Например, благодаря их внедрению в систему грам-
матического разбора, для глагола 'читать' в переходном значении объектом 
может быть только предмет, подлежащий чтению ('книга' или 'журнал', а не 
'обезьяна'). 

Подводя некоторые итоги, отметим, что помимо формально опреде-
ленного набора грамматических правил и словаря, еще одним важнейшим 
компонентом для анализа предложений естественного языка является 
используемая модель ПО, которая служит опорой для проверки семанти-
ческой правильности текста.  Обычно такую проверку  осуществляют, 
непосредственно добавляя логические отношения между формами лексем 
внутри фигурных скобок [2]. Но, в этом случае требуемые для проверки 
лексемы попадают внутрь грамматической сигнатуры, что нарушает 
принцип обобщения. Поэтому, проверку допустимых по смыслу лексем 
следует осуществлять с помощью смысловых агентов, связывающих ана-
лизируемые предложения с отдельно заданной моделью ПО, которая в 
своих структурах содержит атомы, соответствующие базовым формам тех 
или иных лексем, а грамматическое согласование осуществляется путем 
унификации атрибутов лексем, хранимых в словаре. Базовые формы 
имеют именительный падеж единственного числа для существительного и 
прилагательного, а для глагола – неопределенную форму. Тогда, для про-
верки семантической правильности предложения, текущую анали-
зируемую лексему следует привести к базовой форме, предусмотренной в 
модели ПО, так как внутри предложения, в общем случае, все лексемы 
представлены в согласованной форме. Попутно заметим, что для преобра-
зования существительных используется дополнительный атрибут GB, ко-
торый сохраняет признак рода в базовой форме. Это необходимо для 
приведения к базовой форме существительных, когда отличается значе-
ние их рода в согласованной форме от рода в базовой форме. Например, 
слово 'обезьяны' относится к среднему роду, а в базовой форме должно 
иметь женский род, единственное число, именительный падеж, которыми 
представлено соответствующее значение атома в структуре модели ПО. В 
то же время, для определителей (прилагательных) дополнительного атри-
бута, хранящего значение рода в базовой форме, не требуется, так как они 
имеют все родовые текстовые формы, а в базовой форме принят мужской 
род в соответствии с родом понятия 'обьект'. 

 Обычно в Прологе модель ПО описывается декларативно-процедур-
ным методом. Например, факты для предложений о гангстере и обезьяне 
выглядят следующим образом: 

 
'объект'(['гангстер','определитель'(['вооруженный','опасный']), 

       'функция'(['грабить'(['банк','инкассатор','казино'])])]). 
'объект'('обезьяна','определитель'(['смешной','рыжий']), 
          'функция'(['держать'(['банан','палка'])])). 

 
Тогда для успешной работы процедур 'агент_ф'(Lv,F,Fo) и 

'агент_с'(Ls,S,F) переходные глаголы 'грабит' и 'держит' предва-
рительно приводятся к неопределенной форме, а существительное 
'банки'  к единственному числу именительного падежа. Существитель-
ное 'обезьяна' также обрабатывается процедурой приведения к базовой 
форме, вызываемой в соответствии с работой обобщенного алгоритма, 
хотя в данном случае такое преобразование приводит к тавтологии, так 
как исходная текстовая форма этого слова соответствует базовой. 

Исходя из описанного подхода к построению грамматической сигна-
туры, словаря и модели ПО, обратимся к разработке ГНС, где все пере-
численные компоненты представлены раздельно. Это позволяет осущест-
влять смысловой анализ таких фраз, которые вписываются в приведенную 
ранее грамматическую схему. Ограниченный объем статьи, разумеется, не 
дает возможности продемонстрировать детальную семантическую про-
верку большого количества предложений, покрываемых выбранной 
грамматической сигнатурой. Но, в проиллюстрированном далее образце 
программы очевиден резерв для расширения пределов как модели ПО, так 
и словаря, что ведет к более развитому диалогу пользователя с машиной. 
В приведенной программе  можно, допустим, предусмотреть углублен-
ный анализ связи слова 'банки' с моделью ПО. Тогда система будет 
точнее реагировать, например, на некоторое предложение о финансисте, 
открывающем банки, и не перепутает его работу с употреблением консер-
вов, если по каким-то причинам понадобится объединить в одну две раз-
ные модели с настолько противоположным смыслом слов при их одина-
ковой текстовой форме. В этом случае смысловой агент, способный к 
углубленному анализу связи предложения с моделью ПО, потребуется, 
скажем, для фразы: 

 
Успешный состоятельный финансист открывает банки. 

 
Приведенный ниже пример вполне допускает необходимое дополне-

ние. 
 

% интерпретатор Amzi!Prolog 7.2.1 
% выполните consult данного файла и проверьте запросы для разбора предложений 
% ?-'предлож'(['смешная','рыжая','обезьяна','держит','банан'],[]). 
% ?-'предлож'(['знаменитый','талантливый','пародист','рассмешил','публику'],[]). 
% ?-'предлож'(['легкомысленный','халатный','охранник','прозевал','грабителя'],[]). 
% ?-'предлож'(['опасный','вооруженный','гангстер','грабит','банки'],[]). 
:- op(500,xfy,:). 
'предлож'-->'фраза_существит'(G,Q,'именит',C,F),'фраза_глаг'(G,Q,_,_,_,_,C,F). 
'фраза_существит'(G,Q,A,C,F)-->[W1],[W2],[W3], 
   {'прилагат_индекс'(W1,G:Q:A,_,I1), 
    'прилагат_индекс'(W2,G:Q:A,_,I2), 
    'существит_индекс'(W3,G:Q:A:C:GB,_,I3), 

    'прилагат_индекс'(_,'мужск':'единств':'именит',K1,I1), 
    'прилагат_индекс'(_,'мужск':'единств':'именит',K2,I2), 
    'существит_индекс'(_,GB:'единств':'именит':C:_,K3,I3), 
    'агент_о'([K1,K2],S),'агент_с'(K3,S,F)}. 
'фраза_существит'(G,Q,A,C,F)-->[W], 
   {'существит_индекс'(W,G:Q:A:C:GB,_,I), 
   'существит_индекс'(_,GB:'единств':'именит':C:_,K,I), 
    ((is_list(F)->'агент_фо'(K,F);'агент_с'(K,S,F)))}. 
'фраза_глаг'(G,Q,_,_,_,_,C,F)-->[W], 
   {'перех_глаг_индекс'(W,G:Q:'лицо'('третье'):_:V:'действит':C,_,I,Lc), 
    'перех_глаг_индекс'(_,'неопфм':'неопфм':'неопфм':'неопфм':V:'неопфм':C,K,I,Lc), 
    'агент_ф'(K,F,Fo)},'фраза_существит'(_,_,'винит',_,Fo). 
'существит_индекс'(W,G:Q:A:C:GB,K,I):-atomic(W),atomic(I),!,fail. 
'существит_индекс'(W,G:Q:A:C:GB,K,I):-(atomic(W)-> 
('словарь_существит'(G:Q:A:C:GB,L),'элемент'('принадлежать'(W,L)),'индекс'(W,L,I)); 
   (atomic(I),atomic(G),atomic(Q),atomic(A),'словарь_существит'(G:Q:A:C:GB,L), 
    'искать_слово'(K,L,I))). 
'прилагат_индекс'(W,G:Q:A,K,I):-atomic(W),atomic(I),!,fail. 
'прилагат_индекс'(W,G:Q:A,K,I):- 
   (atomic(W)->('словарь_прилагат'(G:Q:A,L), 
                'элемент'('принадлежать'(W,L)), 
                'индекс'(W,L,I)); 
   (atomic(I),atomic(G),atomic(Q),atomic(A), 
                'словарь_прилагат'(G:Q:A,L), 
                'искать_слово'(K,L,I))). 
'перех_глаг_индекс'(W,G:Q:R:T:V:D:C,K,I):-atomic(W), 
   atomic(I),!,fail. 
'перех_глаг_индекс'(W,G:Q:R:T:V:D:C,K,I,Lc):- 
   (atomic(W)->('перех_глаг'(G:Q:R:T:V:D:C,L,Lc), 
                'элемент'('принадлежать'(W,L)), 
                'индекс'(W,L,I)); 
   (atomic(I),atomic(G),atomic(Q),atomic(R), 
    atomic(T),atomic(V),atomic(D),atomic(C), 
                'перех_глаг'(G:Q:R:T:V:D:C,L,Lc), 
                'искать_слово'(K,L,I))). 
'элемент'( P ) :- P, !. 
'подмножество'(L,[]). 
'подмножество'(L,[X|T]):-'элемент'('принадлежать'(X,L)),'подмножество'(L,T). 
'принадлежать'( X, [ X|_ ] ). 
'принадлежать'( X, [ _|T ] ) :- 'принадлежать'( X, T ). 
'индекс'( X, [ ], I ) :- I is 0. 
'индекс'( X, [ X|T ], I ) :- !, I is 1. 
'индекс'( X, [ _|T ], I ) :- 'индекс'( X, T, Ic ),(Ic =:= 0->I is Ic;I is Ic + 1). 
'искать_слово'( X, [ _|_ ], 0 ):- !, fail. 
'искать_слово'( X, [ X|_ ], 1 ). 
'искать_слово'( X, [ _|T ], I ) :- Ic is I - 1,'искать_слово'( X, T, Ic ).  
'агент_о'([K1,K2],S):- Obj='объект'(N,'определитель'(L),F), 
   bagof(Obj,(Obj,'подмножество'(L,[K1,K2])),S). 
'агент_с'(Ls,[],_):-!,fail. 
'агент_с'(Ls,S,F):-'функция_субъект'(Ls,S,[],F). 
'функция_субъект'(Ls,[],F,F). 
'функция_субъект'(Ls,['объект'(Ls,_,'функция'([Fs|_]))|St],Ft,F):- 
      'функция_субъект'(Ls,St,[Fs|Ft],F). 

'агент_ф'(Lv,F,Fo):-'аргум_функция_субъект'(Lv,F,Fo). 
'аргум_функция_субъект'(Lv,[],_):-!,fail. 
'аргум_функция_субъект'(Lv,Lf,Fo):- Lf=[Lfh|Lft], 
   Lfh=..[Fn,Fo],(Lv=Fn->!;'аргум_функция_субъект'(Lv,Lft,Fo)). 
'агент_фо'(Lo,Fo):-'элемент'('принадлежать'(Lo,Fo)). 
'объект'('гангстер','определитель'(['вооруженный','опасный']), 
         'функция'(['грабить'(['банк','инкассатор','казино'])])). 
'объект'('обезьяна','определитель'(['смешной','рыжий']), 
         'функция'(['держать'(['банан','палка'])])). 
'объект'('пародист','определитель'(['знаменитый','талантливый']), 
         'функция'(['рассмешить'(['публика'])])). 
'объект'('охранник','определитель'(['легкомысленный', 
         'халатный']),'функция'(['прозевать'(['грабитель'])])). 
'словарь_существит'('мужск':'единств':'именит':'люди':'мужск', 
   ['пародист','охранник','гангстер','грабитель']). 
'словарь_существит'('мужск':'единств':'родит':'люди':'мужск', 
   ['пародиста','охранника','гангстера','грабителя']). 
'словарь_существит'('мужск':'единств':'дател':'люди':'мужск', 
   ['пародисту','охраннику','гангстеру','грабителю']). 
'словарь_существит'('мужск':'единств':'винит':'люди':'мужск', 
   ['пародиста','охранника','гангстера','грабителя']). 
'словарь_существит'('мужск':'единств':'творит':'люди':'мужск', 
   ['пародистом','охранником','гангстером','грабителем']). 
'словарь_существит'('мужск':'единств':'предлож':'люди':'мужск', 
   ['пародисте','охраннике','гангстере','грабителе']). 
'словарь_существит'('женск':'единств':'именит':'люди':'женск',['публика']). 
'словарь_существит'('женск':'единств':'родит':'люди':'женск',['публики']). 
'словарь_существит'('женск':'единств':'дател':'люди':'женск',['публике']). 
'словарь_существит'('женск':'единств':'винит':'люди':'женск',['публику']). 
'словарь_существит'('женск':'единств':'творит':'люди':'женск',['публикой']). 
'словарь_существит'('женск':'единств':'предлож':'люди':'женск',['публике']). 
'словарь_существит'('женск':'единств':'именит':'животные':'женск',['обезьяна']). 
'словарь_существит'('женск':'единств':'родит':'животные':'женск',['обезьяны']). 
'словарь_существит'('женск':'единств':'дател':'животные':'женск',['обезьяне']). 
'словарь_существит'('женск':'единств':'винит':'животные':'женск',['обезьяну']). 
'словарь_существит'('женск':'единств':'творит':'животные':'женск',['обезьяной']). 
'словарь_существит'('женск':'единств':'предлож':'животные':'женск',['обезьяне']). 
'словарь_существит'('мужск':'единств':'именит':'растения':'мужск',['банан']). 
'словарь_существит'('мужск':'единств':'родит':'растения':'мужск',['банана']). 
'словарь_существит'('мужск':'единств':'дател':'растения':'мужск',['банану']). 
'словарь_существит'('мужск':'единств':'винит':'растения':'мужск',['банан']). 
'словарь_существит'('мужск':'единств':'творит':'растения':'мужск',['бананом'] ). 
'словарь_существит'('мужск':'единств':'предлож':'растения':'мужск',['банане']). 
'словарь_существит'('мужск':'единств':'именит':'учереждения':'мужск',['банк']). 
'словарь_существит'('мужск':'единств':'родит':'учереждения':'мужск',['банка']). 
'словарь_существит'('мужск':'единств':'дател':'учереждения':'мужск',['банку']). 
'словарь_существит'('мужск':'единств':'винит':'учереждения':'мужск',['банк']). 
'словарь_существит'('мужск':'единств':'творит':'учереждения':'мужск',['банком'] ). 
'словарь_существит'('мужск':'единств':'предлож':'учереждения':'мужск',['банке']). 
'словарь_существит'('средн':'множеств':'именит':'учереждения':'мужск',['банки']). 
'словарь_существит'('средн':'множеств':'родит':'учереждения':'мужск',['банков']). 
'словарь_существит'('средн':'множеств':'дател':'учереждения':'мужск',['банкам']). 
'словарь_существит'('средн':'множеств':'винит':'учереждения':'мужск',['банки']). 

'словарь_существит'('средн':'множеств':'творит':'учереждения':'мужск',['банками'] ). 
'словарь_существит'('средн':'множеств':'предлож':'учереждения':'мужск',['банках']). 
'словарь_прилагат'('мужск':'единств':'именит',['вооруженный','знаменитый', 
   'легкомысленный','опасный','рыжий','смешной','талантливый','халатный']). 
'словарь_прилагат'('мужск':'единств':'родит',['вооруженного','знаменитого', 
   'легкомысленного','опасного','рыжего','смешного','талантливого','халатного']). 
'словарь_прилагат'('мужск':'единств':'дател',['вооруженному','знаменитому', 
   'легкомысленному','опасному','рыжему','смешному','талантливому','халатному']). 
'словарь_прилагат'('мужск':'единств':'винит',['вооруженного','знаменитого', 
   'легкомысленного','опасного','рыжего','смешного','талантливого','халатного']). 
'словарь_прилагат'('мужск':'единств':'творит',['вооруженным','знаменитым', 
   'легкомысленным','опасным','рыжим','смешным','талантливым','халатным']). 
'словарь_прилагат'('мужск':'единств':'предлож',['вооруженном','знаменитом', 
   'легкомысленном','опасном','рыжем','смешном','талантливом','халатном']). 
'словарь_прилагат'('женск':'единств':'именит',['вооруженная','знаменитая', 
   'легкомысленная','опасная','рыжая','смешная','талантливая','халатная']). 
'словарь_прилагат'('женск':'единств':'родит',['вооруженную','знаменитую', 
   'легкомысленную','опасную','рыжую','смешную','талантливую','халатную']). 
'словарь_прилагат'('женск':'единств':'дател',['вооруженной','знаменитой', 
   'легкомысленной','опасной','рыжей','смешной','талантливой','халатной']). 
'словарь_прилагат'('женск':'единств':'винит',['вооруженную','знаменитую', 
   'легкомысленную','опасную','рыжую','смешную','талантливую','халатную']). 
'словарь_прилагат'('женск':'единств':'творит',['вооруженной','знаменитой', 
   'легкомысленной','опасной','рыжей','смешной','талантливой','халатной']). 
'словарь_прилагат'('женск':'единств':'предлож',['вооруженной','знаменитой', 
   'легкомысленной','опасной','рыжей','смешной','талантливой','халатной']). 
'перех_глаг'('неопфм':'неопфм':'неопфм':'неопфм': 
   'несоверш':'неопфм':'люди',['грабить'],['люди']). 
'перех_глаг'('неопфм':'неопфм':'неопфм':'неопфм': 
   'несоверш':'неопфм':C,['держать','зевать','смешить'],Lc):- 
   ['животные','люди'] = Lc,'элемент'('принадлежать'(C,Lc)). 
'перех_глаг'('неопфм':'неопфм':'неопфм':'неопфм': 
   'соверш':'неопфм':C,['прозевать','продержать','рассмешить'],Lc):- 
   ['животные','люди'] = Lc,'элемент'('принадлежать'(C,Lc)). 
'перех_глаг'('мужск':'единств':'лицо'('третье'):'наст': 
   'несоверш':'действит':'люди',['грабит'],['люди']). 
'перех_глаг'('мужск':'единств':'лицо'('третье'):'наст': 
   'несоверш':'действит':C,['держит','зевает','смешит'],Lc):- 
   ['животные','люди']=Lc,'элемент'('принадлежать'(C,Lc)). 
'перех_глаг'('женск':'единств':'лицо'('третье'):'наст': 
   'несоверш':'действит':C,['держит','зевает','смешит'],Lc):- 
   ['животные','люди']=Lc,'элемент'('принадлежать'(C,Lc)). 
'перех_глаг'('мужск':'единств':'лицо'('третье'):'прош': 
   'соверш':'действит':C,['прозевал','продержал','рассмешил'],Lc ):- 
   ['животные','люди']= Lc,'элемент'('принадлежать'(C,Lc)). 
'перех_глаг'('женск':'единств':'лицо'('третье'):'прош': 
   'соверш':'действит':C, 
   ['прозевала','продержала','рассмешила'],Lc ) :- 
   ['животные','люди']= Lc,'элемент'('принадлежать'(C,Lc)). 

Раздельное представление грамматической сигнатуры и словаря в ГНС 
обеспечивает обращение к различным моделям проблемных областей зна-
ний в рамках ограниченного естественного языка с формализованным 
стилем записи предложений. Это дает возможность пользователю в своих 
запросах придавать предложениям разный смысл путем дополнения или 
замены словаря и модели ПО без изменения грамматических правил, что 
упрощает разработку гибких систем анализа естественного языка.  

Грамматика непосредственных составляющих допускает расширение 
грамматических признаков частей речи смысловыми признаками, кото-
рые посредством связанных переменных Пролога передаются для анализа 
смысловым агентам.  

Внедрение в систему грамматического разбора смысловых агентов по-
зволяет осуществлять проверку не только грамматической, но и смысло-
вой правильности лексем в их контекстной взаимосвязи внутри предло-
жений естественного языка. В этом случае смысловые агенты служат 
каналом обмена семантической информацией между анализируемыми 
предложениями и моделью ПО вдоль всей цепочки грамматического раз-
бора. Это позволяет удерживать смысл анализируемых предложений, в 
пределах ограничений, предусмотренных на сопутствующих грамматиче-
скому разбору структурах модели ПО, которая в целом образует семанти-
ческую опору системы обработки естественного языка.  

Рассмотренная модель ПО охватывает базовые формы всех помещен-
ных в словарь лексем, объединяя их в допустимые смысловые отношения, 
и в типовом случае описывается декларативно-процедурным методом, что 
позволяет адаптировать ГНС к формальному семантическому анализу 
предложений естественного языка в пределах заданных ограничений. 

